
Session 5: Headquarters Management Programs 
 

At the end of this lesson participants will be able to: 

• Implement a basic survey management system to run at headquarters 

 

Structure of the HQ Menu Program 
In addition to the menu programs for the supervisor and the interviewer, we will also implement a 

menu program for managing the survey from the central office. This menu program will allow central 

office staff to manage assignments of clusters to supervisors, download data and view reports. It could 

be run on a desktop computer or a mobile device. 

This menu program will have the following screens: 

 
 

As with our other menu programs, we create a new CAPI application with a new dictionary and add an 

entry in the dictionary for each menu. In this program we will have three items in the dictionary: Assign 

clusters, Choose cluster and Choose supervisor. Sync and cluster status reports will be implemented as 

user functions. After dragging the three items from the dictionary onto the form, we implement the 

main menu as follows: 
 

  

Main menu 

• Assign clusters 

• Sync 

• Cluster status report 

  

Choose cluster 

• <cluster 1> 

• <cluster 2> 

• <cluster 3> 

• … 

• Back 

Choose supervisor 

• <sup 1> 

• <sup 2> 

• <sup 3> 

• … 

• Back 

PROC MAIN_MENU 

onfocus 

$ = notappl; 

 

if $ = 1 then 

    // Assign clusters 

    skip to CHOOSE_CLUSTER_TO_ASSIGN; 

elseif $ = 2 then 

    // Sync 

    synchronize(); 

elseif $ = 3 then 

    // Cluster status report 

    clusterStatusReport(); 

endif; 

 

reenter; 

 

Assigning Clusters 
A major function of our HQ menu program will be to assign clusters to supervisors. In order to 

implement this, we will add a new variable to the listing assignments dictionary to store the staff code of 

the assigned supervisor. We will name it LA_ASSIGNED_SUPERVISOR. Initially it will be blank but will be 

filled in with the supervisor staff code once the cluster is assigned to a supervisor. 

The CHOOSE_CLUSTER_TO_ASSIGN onfocus will show a dynamic value set containing the list of clusters 

that are in the "not started" state. These are the clusters that can be assigned to supervisors. 

PROC CHOOSE_CLUSTER_TO_ASSIGN 

onfocus 

$ = ""; 

 

// Create value set from all clusters with state "not stared" 

numeric nextEntryValueSet = 1; 

forcase LISTINGASSIGNMENTS_DICT where LA_CLUSTER_STATUS = 1 do 

     

    // Get name of currently assigned supervisor from staff file 

    string assignedSupervisorName; 

    if LA_ASSIGNED_SUPERVISOR = notappl then 

        assignedSupervisorName = "unassigned"; 

    else 

        loadcase(STAFF_DICT, LA_ASSIGNED_SUPERVISOR); 

        assignedSupervisorName = STAFF_NAME; 

    endif; 

     

    codesString(nextEntryValueSet) = key(LISTINGASSIGNMENTS_DICT); 

    labels(nextEntryValueSet) = maketext("%v-%v-%v : %s",  

        LA_PROVINCE, LA_DISTRICT, LA_CLUSTER, assignedSupervisorName); 

    nextEntryValueSet = nextEntryValueSet + 1; 

enddo; 

 

codesString(nextEntryValueSet) = "back"; 

labels(nextEntryValueSet) = "Back"; 

nextEntryValueSet = nextEntryValueSet + 1; 

codesString(nextEntryValueSet) = ""; 

 

setvalueset($, codesString, labels); 

 

In the postproc we handle the back option or continue to the next field on the form, 

ASSIGN_CLUSTER_TO. 

postproc 

 

if $ = "back" then 

    reenter MAIN_MENU; 

endif; 

 

In the onfocus of ASSIGN_CLUSTER_TO we show the list of supervisors that the cluster may be assigned 

to by looping through the staff file. 

PROC ASSIGN_CLUSTER_TO 

onfocus 

$ = notappl; 

 

// Create value set of supervisors 

numeric nextEntryValueSet = 1; 

forcase STAFF_DICT where STAFF_ROLE = 2 do 

         

    codes(nextEntryValueSet) = STAFF_CODE; 

    labels(nextEntryValueSet) = STAFF_NAME; 

    nextEntryValueSet = nextEntryValueSet + 1; 

enddo; 

 

codes(nextEntryValueSet) = -1; 

labels(nextEntryValueSet) = "Back"; 

nextEntryValueSet = nextEntryValueSet + 1; 

 

codes(nextEntryValueSet) = notappl; 

 

setvalueset($, codes, labels); 

In the postproc, we set the assigned supervisor to the chosen staff member and write the case. 

postproc 

 

if $ = -1 then 

    reenter CHOOSE_CLUSTER_TO_ASSIGN; 

endif; 

 

LA_PROVINCE = tonumber(CHOOSE_CLUSTER_TO_ASSIGN[1:1]); 

LA_DISTRICT = tonumber(CHOOSE_CLUSTER_TO_ASSIGN[2:2]); 

LA_CLUSTER = tonumber(CHOOSE_CLUSTER_TO_ASSIGN[4:3]); 

loadcase(LISTINGASSIGNMENTS_DICT, LA_PROVINCE, LA_DISTRICT, LA_CLUSTER); 

LA_ASSIGNED_SUPERVISOR = $; 

LA_ASSIGNED_INTERVIEWER = notappl; 

LA_CLUSTER_STATUS = 1; 

writecase(LISTINGASSIGNMENTS_DICT); 

reenter CHOOSE_CLUSTER_TO_ASSIGN; 

 

All that is left now is to go back and modify the supervisor menu program so that the supervisor may 

only assign clusters for listing that are already assigned to her.  

PROC CHOOSE_CLUSTER_TO_ASSIGN 

onfocus 

$ = ""; 

 

// Create dynamic value set from listing 

// assignment file, only includes clusters that are 

// not started or currently assigned for listing 

numeric nextEntryValueSet = 1; 

forcase LISTINGASSIGNMENTS_DICT where LA_CLUSTER_STATUS in 1:2  

    and LA_ASSIGNED_SUPERVISOR = LOGIN do 

 

Synchronizing 
The key data to synchronize is the listing assignments file. It needs to be synced both ways in order to 

send the new assignments made at the central office to devices in the field as well as to receive the 

latest cluster statuses from the field. In addition to syncing the listing assignments file we will also 

download the latest household and listing data files for use in reports. 

function synchronize() 

    if syncconnect(CSWeb, "http://csweb.teleyah.com/api") then 

         

        // Bi-directional sync of listing assignments 

        syncdata(BOTH, LISTINGASSIGNMENTS_DICT); 

         

        // Download data files 

        syncdata(GET, HOUSEHOLD_DICT); 

        syncdata(GET, LISTING_DICT); 

         

    endif; 

end; 

Reports 
The central office management menu will also include reports. Just like in the supervisor menu, we will 

use HTML templates to create reports. The cluster status report will show the progress for each cluster 

including the status, assigned supervisor, number of listed households and number of interviewed 

households. 

 

We will use a multiply occurring record in the working storage dictionary to store the data for the table.  

Each occurrence of the record will store data for one row in the table. Add a working storage dictionary 

to the application and in it add a record named WS_CLUSTER_STATUS_REC with 999 occurrences 

(enough to hold all the clusters in our survey). Add the following items to this record: 

• WS_CLUSTER_STATUS_PROVINCE (numeric, length 1) 

• WS_CLUSTER_STATUS_DISTRICT (numeric, length 2) 

• WS_CLUSTER_STATUS_CLUSTER (numeric, length 3) 

• WS_CLUSTER_STATUS_STATUS (alpha, length 30) 

• WS_CLUSTER_STATUS_SUPERVISOR (alpha, length 30) 

• WS_CLUSTER_STATUS_LISTED (numeric, length 3) 

• WS_CLUSTER_STATUS_INTERVIEWED (numeric, length 3) 

Our clusterStatusReport function will fill in the occurrences of this record and run the report. The first 

step is to clear the extra occurrences from the WS_CLUSTER_STATUS record so that we can add in the 

correct number. 

function clusterStatusReport() 

 

    setreportdata(clear); 

    clear(WS_DICT); 

     

    // Remove default occurrences of record 

    do numeric i = count(WS_CLUSTER_STATUS_REC) until i <= 0 by (-1) 

        delete (WS_CLUSTER_STATUS_REC(i)); 

    enddo; 

 

Next, we loop through all the clusters in the listing assignments dictionary and add a new occurrence for 

each cluster. 

    numeric i = 1; 

    forcase LISTINGASSIGNMENTS_DICT do 

        insert(WS_CLUSTER_STATUS_REC(i)); 

Then we copy the cluster ids from the listing assignments dictionary into the working storage record. 

        WS_CLUSTER_STATUS_PROVINCE(i) = LA_PROVINCE; 

        WS_CLUSTER_STATUS_DISTRICT(i) = LA_DISTRICT; 

        WS_CLUSTER_STATUS_CLUSTER(i) = LA_CLUSTER; 

 

We can use getlabel() to get the a status of the cluster as a string instead of a number and add that to 

the working storage record. 

        WS_CLUSTER_STATUS_STATUS(i) =  

                      getlabel(LA_CLUSTER_STATUS_VS1, LA_CLUSTER_STATUS); 

 

To get the supervisor name we use loadcase to look it up in the staff file based on the supervisor code in 

the listing assignments file. 

        if LA_ASSIGNED_SUPERVISOR <> notappl then 

            loadcase(STAFF_DICT, LA_ASSIGNED_SUPERVISOR); 

            WS_CLUSTER_STATUS_SUPERVISOR(i) = STAFF_NAME; 

        else 

            WS_CLUSTER_STATUS_SUPERVISOR(i) = ""; 

        endif; 

 

The total number of listed households in the cluster is obtained by loading the cluster from the listing 

file and using count to get the number of rows in the roster. 

        if loadcase(LISTING_DICT, LA_PROVINCE, LA_DISTRICT, LA_CLUSTER) then 

            WS_CLUSTER_STATUS_LISTED(i) = count(LISTING_REC); 

        else 

            WS_CLUSTER_STATUS_LISTED(i) = 0; 

        endif; 

 

We compute the number of interviewed households by looping through the household data file to get 

all the completed (not partial) cases in the cluster. Normally we would use countcases() to do this but 

countcases() cannot be used inside a forcase loop, nor can another forcase loop. To work around this, 

we use locate and loadcase. 

        WS_CLUSTER_STATUS_INTERVIEWED(i) = 0; 

        locate(HOUSEHOLD_DICT, >=, ""); 

        while loadcase(HOUSEHOLD_DICT) do 

            if PROVINCE = LA_PROVINCE and DISTRICT = LA_DISTRICT and 

                    CLUSTER = LA_CLUSTER and  

                    not ispartial(HOUSEHOLD_DICT) then 

                inc(WS_CLUSTER_STATUS_INTERVIEWED(i)); 

            endif; 

        enddo; 

 

Finally we end the loop, attach the working storage to the report using setreportdata() and run the 

report. 

        inc(i); 

    endfor; 

     

    setreportdata(WS_CLUSTER_STATUS_REC); 

    report("../Reports/clusterstatus.html"); 

end; 

 

The template for our report is just and HTML table with columns matching the variables in the working 

storage record. 

<!doctype html> 

<html lang="en"> 

<head> 

    <meta charset="utf-8"> 

    <meta name="viewport" content="width=device-width, initial-scale=1, shrink-to-fit=no"> 

    <link rel="stylesheet" href="css/bootstrap.min.css"> 

    <title>Cluster Status Report</title> 

</head> 

<body> 

    <h1>Cluster Status Report</h1> 

    <br/> 

    <div id="cspro_report"><!-- replaced by expanded template --></div> 

 

    <!-- report template --> 

    <script type="application/vnd.cspro.report-template" id="cspro_report_template"> 

        <table class="table table-bordered"> 

                <thead> 

                    <tr> 

                        <th>Province</th> 

                        <th>District</th> 

                        <th>Cluster</th> 

                        <th>Status</th> 

                        <th>Supervisor</th> 

                        <th>Listed</th> 

                        <th>Interviewed</th> 

                    </tr> 

                </thead> 

                <tbody> 

                    {{#WS_CLUSTER_STATUS_REC}} 

                    <tr> 

                        <td>{{WS_CLUSTER_STATUS_PROVINCE}}</td> 

                        <td>{{WS_CLUSTER_STATUS_DISTRICT}}</td> 

                        <td>{{WS_CLUSTER_STATUS_CLUSTER}}</td> 

                        <td>{{WS_CLUSTER_STATUS_STATUS}}</td> 

                        <td>{{WS_CLUSTER_STATUS_SUPERVISOR}}</td> 

                        <td>{{WS_CLUSTER_STATUS_LISTED}}</td> 

                        <td>{{WS_CLUSTER_STATUS_INTERVIEWED}}</td> 

                    </tr> 

                    {{/WS_CLUSTER_STATUS_REC}} 

                </tbody> 

            </table> 

    </script> 

    <script src="js/jquery-3.3.1.slim.min.js"></script> 

    <script src="js/popper.min.js"></script> 

    <script src="js/bootstrap.min.js"></script> 

</body> 

</html> 

 

While our HQ menu program is now fully functional, it is far from complete. We should add additional 

reports as well functions to manage the staff data file by adding new staff members and assigning 

interviewers to supervisors. This is left as an exercise for the reader. 


